Two intervals with Dirichlet BCs

AJR

June 30, 2015

Let's try the simplest problem. The finite domain is $-1 \leq x \leq 1$, t is time, and we seek a field $u(x,t)$ satisfying the heat/Burgers' PDE $u_t = u_{xx} - \alpha u u_x$ and Dirichlet physical boundary conditions of $u = 0$ at $x = \pm 1$. The simplest numerical approximation in our class is formed by imposing the artificial breakpoint at $x = 0$ that the field is continuous, $[u]_0 = 0$, but the derivative has a jump of $[u_x]_0 = (1-\gamma)(0 - 2u|_0 + 0) = -2(1-\gamma)u|_0$.

1 Initialisation

```
on div; off allfac; on revpri;
factor uu,alpha;
```

The slow manifold depends upon `uu` $= U(t) := u(0,t)$. Its evolution is $dU/dt = \mathtt{g}$.

```
depend uu,t;
let df(uu,t)=>g;
```

Slow subspace approximation to the slow manifold.

```
u:=(1-xx)*uu;
g:=0;
```

The subgrid field is expressed in terms of `xx` $:= |x|$ so define `sx` $:= \operatorname{sign} x$:

```
depend xx,x;
```

```
let { df(xx,x)=>sx, sx^2=>1 };
```

```
operator iint;linear iint;
let { iint(1,x)=>(xx^2-xx)/2
    , iint(xx^~~p,x)=> (xx^(p+2)-xx)/(p+1)/(p+2) };
```

2 Iterative construction

```
let { gamma^40=>0 };
alpha:=0*alfa*gamma;
aa:=0;% 1/6 may be useful for low orders, but only a little at hi
for it:=1:99 do begin
```

Compute the residual of the PDE and coupling condition.

```
respde:=-df(u,t)+df(u,x,x)-alpha*u*df(u,x);
ux:=df(u,x);
rescc:=sub(xx=0,(1-aa*gamma)*(sub(sx=1,ux)-sub(sx=-1,ux))
    +2*(1-gamma)*u);
```

To monitor progress, write the lengths of the residual expressions:

```
write lengthress:={length(respde),length(rescc)};
```

Compute corrections from residuals.

```
g:=g+(gd:=3/2*rescc+3*sub({xx=1,sx=0},int(respde*(1-xx),xx)));
u:=u+iint(gd*(1-xx)-respde,x);
```

Terminate loop when residuals are zero.

```
if {respde,rescc}={0,0} then write "success ",it:=10000+it;
end;
```

3 Generalised Domb–Sykes plot

If linear diffusion and to high order, then post-process to generate a generalised Domb–Sykes plot (Mercer & Roberts, 1990).

```
if alpha=0 and deg((1+gamma)^20,gamma)>19 then begin
on rounded; print_precision 4;
as:=coeff(g,gamma);
write
rks:=for k:=5:length(as)-2 collect 1/(k-1);
write
bs:=for k:=5:length(as)-2 collect sqrt(
    (part(as,k+2)*part(as,k)-part(as,k+1)^2)
   /(part(as,k+1)*part(as,k-1)-part(as,k)^2) );
write
cs:=for k:=5:length(as)-2 collect (
    part(as,k)*part(bs,k-4)+part(as,k+2)/part(bs,k-4)
    )/part(as,k+1)/2;
out "tiwdbcDS.tex";
write "
\begin{tikzpicture}
\begin{axis}[axis x line=middle, axis y line=middle
,xmin=0,ymax=0.27]
\addplot +[only marks,mark=*] coordinates {"$
for k:=1:length(rks) do write "(",part(rks,k),",",part(bs,k),")"
write "};
\addlegendentry{\(B_k\) vs \(1/k\)};"$
write "
\addplot +[only marks,mark=*] coordinates {"$
for k:=1:length(rks) do write "(",part(rks,k)^2,",",-part(cs,k),
write "};
\addlegendentry{\(-\cos\theta_k\) vs \(1/k^2\)};
\end{axis}
\end{tikzpicture}
"$
```



Figure 1: generalised Domb–Sykes plot (Mercer & Roberts, 1990) to show convergence for $|\gamma| < 3.8$ due to a convergence limiting singularity at angle 103° to the real γ-axis.

```
shut "tiwdbcDS.tex";
end;
```

Fin.

```
end;
```