Two intervals with Dirichlet BCs

AJR

June 16, 2015

Let's try the simplest problem. The finite domain is $-1 \le x \le 1$, t is time, and we seek a field $u(x,t)$ satisfying the heat/Burgers' PDE $u_t = u_{xx} - \alpha u u_x$ and Dirichlet physical boundary conditions of $u = 0$ at $x = \pm 1$. The simplest numerical approximation in our class is formed by imposing the artificial breakpoint at $x = 0$ that the field is continuous, $[u]_0 = 0$, but the derivative has a jump of $[u_x]_0 = (1-\gamma)(0 - 2u|_0 + 0) = -2(1-\gamma)u|_0$.

```
on div; off allfac; on revpri;
factor uu,alpha;
```

The slow manifold depends upon `uu` $= U(t) := u(0,t)$. Its evolution is $dU/dt = \mathtt{g}$.

```
depend uu,t;
let df(uu,t)=>g;
```

Slow subspace approximation to the slow manifold.

```
u:=(1-xx)*uu;
g:=0;
```

The subgrid field is expressed in terms of `xx` $:= |x|$ so define `sx` $:= \operatorname{sign} x$:

```
depend xx,x;
let { df(xx,x)=>sx, sx^2=>1 };
```

```
operator iint;linear iint;
let { iint(1,x)=>(xx^2-xx)/2
```

```
    , iint(xx^~~p,x)=> (xx^(p+2)-xx)/(p+1)/(p+2) };
let { gamma^3=>0, alpha=>0};
for it:=1:9 do begin
```

Compute the residual of the PDE and coupling condition.

```
write
respde:=-df(u,t)+df(u,x,x)-alpha*u*df(u,x);
ux:=df(u,x);
write
rescc:=sub(xx=0,sub(sx=1,ux)-sub(sx=-1,ux)+2*(1-gamma)*u);
```

Compute corrections from residuals.

```
g:=g+(gd:=3/2*rescc+3*sub({xx=1,sx=0},int(respde*(1-xx),xx)));
u:=u+iint(gd*(1-xx)-respde,x);
```

Terminate loop when residuals are zero.

```
if {respde,rescc}={0,0} then write "success ",it:=10000+it;
end;

end;
```