

List of pdfcomments

AJR - This last potential should have the advantage that subgrid fields are entirely polynomial, albeit involving $ x $. The potential is normalised to have the same height as for Alfimov et al. (2002), although the widths of the wells are different.	3
---	---

Discretisation of nonlinear Schrodinger PDE with periodic potential

AJR

January 28, 2014

Contents


List of pdfcomments	1
1 Introduction	3
2 Computer algebra construction	4
2.1 Subgrid variable	7
2.2 Operators to find updates to approximations	7
2.3 Initialise the slow manifold	8
2.4 Iterate to satisfy nlS PDE and coupling	9
3 Post-processing	11
3.1 Check continuity	11
3.2 Equivalent differential equation maybe	12
3.3 Optionally plot subgrid fields	12
3.4 Try to match with Wannier results	13

1 Introduction

The immediate aim is to model the dynamics in just one space dimension of the nonlinear Schrodinger PDE (1) with periodic potential. That is, we seek to model the field $u(x, t)$ that solves

$$-i \frac{\partial u}{\partial t} = \frac{\partial^2 u}{\partial x^2} - V(x)u - \sigma |u|^2 u \quad (1)$$

where here, for example, we could take potential

- $V(x) := \nu[1 - \cos(2\pi x/H)]\pi^2/2/H^2$,
- or $V(x) := A \cos(2\pi x/H)\pi^2/H^2$ to agree with [Alfimov et al. \(2002\)](#) when $H = \pi$ (but we all use negative A to get localisation about positions $x = jH$),
- or perhaps $V(x) := B[1 - x^2(|x| - H)^2 32/H^4]\pi^2/H^2$ for $|x| < H$ and periodically extended. ($A \approx B \approx -\nu/2$ should be much the same.) Can find to errors $\mathcal{O}(\gamma^4, B^5)$ in about 20 seconds. 
- We could even multiply this last potential by the factor

$$\left[\frac{2}{3} + \frac{1}{3}x^2(|x| - H)^2 16/H^2\right]$$

so that it looks more like the cosine (maybe about 5% error).

The plan is to have ‘amplitudes’ that measures what goes on in the j th well which is centred about $X_j = jH$. To do this we notionally divide space into overlapping elements; the j th element being $E_j := [X_{j-1}, X_{j+1}]$ which is centred on X_j . The subgrid field in each element, $u_j(x, t)$, is coupled to its neighbours by

$$u_j(X_{j\pm 1}, t) = \gamma u_{j\pm 1}(X_{j\pm 1}, t) + (1 - \gamma)u_j(X_j, t), \quad (2)$$

Then we construct a model in powers of γ that parametrises the coupling between elements, and hence the coupling between wells. For simplicity, we

construct the model in powers of the strength of the well because that is an easy thing to do: quick and dirty.

For example, and following [Alfimov et al. \(2002\)](#), for spacing $H = \pi$, this algorithm constructs the model of linear dynamics, $\sigma = 0$, that

$$-i\dot{U}_j = -(\frac{1}{2}\nu - \frac{1}{32}\nu^2)U_j + \gamma(1 - \frac{1}{32}\nu^2)\frac{1}{\pi^2}(U_{j-1} - 2U_j + U_{j+1}) + \mathcal{O}(\gamma^2, \nu^3, \sigma). \quad (3)$$

The subgrid fields of the slow manifold are complicated, even at this low order of truncation: the first few terms are

$$u_j = U_j + \frac{1}{8}\nu \cos 2\theta U_j + \gamma \left[\frac{1}{\pi}\theta\mu\delta + (\frac{1}{2\pi^2}\theta^2 - \frac{1}{24})\delta^2 \right] U_j + \mathcal{O}(\gamma^2 + \nu^2), \quad (4)$$

for centred mean and difference operators, μ and δ . Figure 1 plots an example: there appears to be little penetration through the potential barriers.

The evolution (3) shows two effects:

- the term starting $-\frac{1}{2}\nu U_j$ changes the frequency of solutions on the slow manifold as the potential barriers start to grow (presumably we could change the potential $V(x)$ so this detuning was zero, if we wish);
- the coefficient $\frac{1}{\pi^2}(1 - \frac{1}{32}\nu^2)$ of the discrete diffusion $\delta^2 U_j$ confirms that increasing potential barriers decrease the communication (tunnelling) between the wells.

We should be able to compare these results to those of [Alfimov et al. \(2002\)](#).

Theory Should have a section on invariant manifold theory that supports the slow manifold as some coarse model of the PDE.

2 Computer algebra construction

The computer algebra code uses Reduce, available freely.¹

¹<http://www.reduce-algebra.com>




Figure 1: slow manifold subgrid fields for $U_0 = 1$ and all other $U_j = 0$, evaluated at potential strength $\nu = 3$, for the approximation with errors $\mathcal{O}(\gamma^3, \nu^3, \sigma)$: red, $u_0(\theta)$ on $E_0 = [-\pi, \pi]$; green, $u_1(\theta)$ on $E_1 = [0, 2\pi]$; blue, $u_2(\theta)$ on $E_2 = [\pi, 3\pi]$.

Execute with `in_tex "nls.tex"$`

Improve appearance of printing.

```
1 on div; on revpri; off allfac;
2 factor hh,i,nu,aa,bb,sigma,pi,df;
```

The following empowers complex conjugation operator so we just solve one nls PDE.

```
3 operator cc;
4 let { cc(~u*~v)=>cc(u)*cc(v)
5      , cc(~u/~v)=>cc(u)/cc(v)
6      , cc(~u+~v)=>cc(u)+cc(v)
7      , cc(~u^~p)=>cc(u)^p
8      , df(cc(~v),~u)=>cc(df(v,u))
9      , cc(i)=>-i, cc(-i)=>i
10     , cc(~u)=>u when numberp(u)
11     , cc(cc(~u))=>u
12     , cc(q)=>q
13     , cc(cos(~u))=>cos(u)
14     , cc(sin(~u))=>sin(u)
15     , cc(sign(~u))=>sign(u)
16     , cc(nu)=>nu
17     , cc(aa)=>aa
18     , cc(bb)=>bb
19     , cc(sigma)=>sigma
20     , cc(gamma)=>gamma
21     , cc(pi)=>pi
22     , cc(hh)=>hh
23   };
```

The following empowers using the sign function to get dependence upon $|\theta|$:

it transforms all high powers to just the first or the zeroth; and its derivative is zero upon ignoring the possible delta-function.

```
24 let { sign(~u)^2=>1
25      , df(sign(~u),~v)=>0 };
```

2.1 Subgrid variable

Make subgrid structures a function of element phase $\theta = \pi(x - X_j)/H$ for element size H ; denote phase θ by **q**. Observe: the phase θ changes by 2π over one overlapping element; the boundaries of the j th element are at the adjacent grid points at $\theta = \pm\pi$.

```
26 depend q,x;
27 let df(q,x)=>pi/hh;
```

2.2 Operators to find updates to approximations

These ‘quick and dirty’ linear operators are not the best, but they are good enough to achieve the aim of satisfying the PDE and coupling conditions.

Procedure **mean** computes the mean over the j th element, precisely $\text{mean}(f) := \frac{1}{H} \int_{-H/2}^{H/2} f dx$: it finds solvability conditions; and is currently used for the amplitude.

```
28 operator mean; linear mean;
29 let { mean(1,q)=>1
30      , mean(q^~p,q)=>(pi/2)^p*(1+(-1)^p)/2/(p+1)
31      , mean(sign(q),q)=>0
32      , mean(sign(q)*q^~p,q)=>(pi/2)^p*(1-(-1)^p)/2/(p+1)}
```

```

33      , mean(cos(~m*q),q)=>2*sin(m*pi/2)/m/pi
34      , mean(sin(~a),q)=>0
35      , mean(q^~p*cos(~m*q),q)=>(
36        +(pi/2)^(p-1)*sin(m*pi/2)*(1+(-1)^p)/2
37        -p*mean(q^(p-1)*sin(m*q),q) )/m
38      , mean(q^~p*sin(~m*q),q)=>(
39        -(pi/2)^(p-1)*cos(m*pi/2)*(1-(-1)^p)/2
40        +p*mean(q^(p-1)*cos(m*q),q) )/m
41    };

```

The linear operator `solv` used above solves $\mathcal{L}u = -\frac{\pi^2}{H^2}u_{\theta\theta} = \text{RHS}$ such that $u(0, t) = 0$ and $u(\pi, t) = u(-\pi, t)$.

```

42 operator solv; linear solv;
43 let { solv(q^~p,q)=>(hh/pi)^2*( -q^(p+2)
44      +q*pi^(p+1)*(1-(-1)^p)/2 )/(p+2)/(p+1)
45      , solv(1,q)=>(hh/pi)^2*(-q^2)/2
46      , solv(sign(q)*q^~p,q)=>(hh/pi)^2*( -q^(p+2)*sign(q)
47      +q*pi^(p+1)*(1+(-1)^p)/2 )/(p+2)/(p+1)
48      , solv(sign(q),q)=>(hh/pi)^2*sign(q)*(-q^2)/2
49      , solv(cos(~m*q),q)=>(cos(m*q)-1)*(hh/pi/m)^2
50      , solv(q^~p*cos(~m*q),q)=>q^p*(cos(m*q)-1)*(hh/pi/m)^2
51      , solv(sin(~m*q),q)=>sin(m*q)*(hh/pi/m)^2
52      , solv(q^~p*sin(~m*q),q)=>q^p*sin(m*q)*(hh/pi/m)^2
53    };

```

2.3 Initialise the slow manifold

The slow manifold is that the subgrid field depends upon the evolving amplitude such as $U_j(t) := u_j(X_j, t)$ or currently the average $U_j(t) := \frac{1}{H} \int_{-H/2}^{H/2} u_j(x, t) dx$.


```

54 operator uu; depend uu,t;
55 let df(uu(~k),t)=>sub(j=k,gj) ;

```

The initial subgrid field and evolution is the subspace of piecewise constant fields. This code only generates the slow manifold tangent to this slow subspace: interactions between multiple modes in the same potential well will need significantly extended code.

```

56 uj:=uu(j); gj:=0;

```

2.4 Iterate to satisfy nLS PDE and coupling

Iterate in a loop until residuals are zero to specified order of error. The independent small parameters are:

- γ , parametrises the inter-element coupling;
- ν , A or B , the strength of the potential wells;
- σ , the strength of the nonlinearity.

We will probably link some of these small parameters at sometime.

Set an Euler transform parameter ([Van Dyke 1964](#), e.g.), probably should depend upon potential strength, but do not yet know how. For $A = B = -1$ try

```

57 Eu:=1/3;

58 let { gamma^7=>0, sigma=>0, nu=>0, aa=>0, bb^3=>0 };
59 for it:=1:99 do begin

```

Compute the residual of the PDE (1) and coupling conditions (2): these drive updates to the approximations. Also trace print the algebraic length of the residuals so we can see how the iteration is proceeding.

```

60 %write
61   potl:=( nu*(1-cos(2*q))/2
62           +aa*cos(2*q)
63           +bb*(1-32/pi^4*q^2*(q*sign(q)-pi)^2)
64           )*pi^2/hh^2;
65 %write
66   upde:=trigsimp(
67     +i*df(uj,t)+df(uj,x,x)-potl*uj-sigma*cc(uj)*uj^2
68     ,combine);
69 %write
70   urcc:=(1+Eu/(1-Eu)*gamma)*sub(q=+pi,uj)
71     -gamma/(1-Eu)*sub({q=0,j=j+1},uj)
72     -(1-gamma)*sub(q=0,uj);
73 %write
74   ulcc:=(1+Eu/(1-Eu)*gamma)*sub(q=-pi,uj)
75     -gamma/(1-Eu)*sub({q=0,j=j-1},uj)
76     -(1-gamma)*sub(q=0,uj);
77   write lengthResiduals:=map(length(~a)
78     ,{upde,urcc,ulcc,uamp});

```

Use the defined linear operators to update the approximate slow manifold subgrid field and evolution.

```

79 %write
80   gj:=gj+i*(gd:=mean(upde,q)-(urcc+ulcc)/hh^2);
81 %write
82   uj:=uj+solv(upde-gd,q)+q*(-urcc+ulcc)/2/pi;

```

Fix the amplitude: although better to do this in `solv`, to be flexible we can do it here. This code fixes U_j to be the mean over non-overlapping elements.

```
83 %write
84   uj:=uj-(uamp:=mean(uj,q)-uu(j));
```

Terminate the iteration when all residuals are zero, to specified error, and print an information number.

```
85   showtime;
86   if {upde,urcc,ulcc,uamp}={0,0,0,0}
87   then write it:=it+100000;
88 end;
```

3 Post-processing

3.1 Check continuity

The field and its first four derivatives are continuous, but not the fifth derivative in the case of the polynomial potential.

```
89 vj:=sub(sign(q)=1,uj)-sub(sign(q)=-1,uj)$
90 cty0:=sub(q=0,vj);
91 cty1:=sub(q=0,(vj:=df(vj,x)));
92 cty2:=sub(q=0,(vj:=df(vj,x)));
93 cty3:=sub(q=0,(vj:=df(vj,x)));
94 cty4:=sub(q=0,(vj:=df(vj,x)));
95 cty5:=length(sub(q=0,(vj:=df(vj,x))));
```

3.2 Equivalent differential equation maybe

Determine the equivalent differential equation for amplitudes that vary slowly over the wells.

```

96 if 0 then begin
97   let hh^9=>0;
98   depend uu,x; depend vv,x;
99   taylor:={ uu(j)=>uu
100      , uu(j+~p)=>uu+(for n:=1:9 sum
101      df(uu,x,n)*(hh*p)^n/factorial(n))
102      }$
103   migde:=(-i*gj where taylor);
104 end;
```

3.3 Optionally plot subgrid fields

For simplicity let's set $H = \pi$, but the code works for general H .

```

105 hh:=pi;
```

Optionally plot some fully coupled subgrid fields of the linear problem, $\sigma = 0$, for a potential strength $\nu = 3$, say. Or set $A, B \in \{-1, -5, -15\}$ to compare with [Alfimov et al. \(2002\)](#). For expressions with many terms, it would be quicker to output to a file and draw graph in Matlab/Octave/Scilab (I have a bash script that would help edit).

```

106 load_package gnuplot;
107 % length less than 300 is easy enough to plot
108 write length(uj);
109 if length(uj)<300 then begin
```

```

110 hh:=pi;
111 gamma:=1; sigma:=0; nu:=3; aa:=-1; bb:=-1;
112 uj0:=coeffn(uj,uu(j),1)$
113 uj1:=sub(q=q-pi,coeffn(uj,uu(j-1),1))$
114 uj2:=sub(q=q-2*pi,coeffn(uj,uu(j-2),1))$
115 ujs:=map(max(-1,min(2,~a)),{uj0,uj1,uj2});
116 plot(ujs,q=(-pi .. 3*pi));
117 end;

```

3.4 Try to match with Wannier results

What does the interaction look like for specific values? Seems to agree moderately well with first column of Table I of [Alfimov et al. \(2002\)](#), but as yet unclear if the differences will go to zero or not as higher order terms are computed.

```

118 on rounded; print_precision 4$
119 gamma:=1; sigma:=0; nu:=3; aa:=-1; bb:=-1;
120 idUdt:=i*gj;
121 clear gamma; clear aa; clear bb;
122 hatw01:=coeffn(i*gj,uu(j),1);
123 hatw11:=coeffn(i*gj,uu(j+1),1);

```

In these $\hat{\omega}_{n,\alpha}$: $\mathcal{O}(A^2)$ coefficients are mostly 0.01–0.02; $\mathcal{O}(A^4)$ coefficients are 0.002–0.003. Similarly for quadratic potential parametrised by B .

But the convergence of the coefficients in γ appears quite slow, the terms decay maybe like $(2/3)^n \gamma^n$. Suggest may be a convergence limiting singularity at $\gamma \approx +1.5$. Could try an Euler transform, $\gamma = \gamma'/(1 - E + E\gamma')$ equivalently $\gamma' = (1 - E)\gamma/(1 - E\gamma)$ for say $E = \frac{2}{3}$ or a bit more conservatively $E = \frac{1}{2}$. At $E = \frac{1}{2}$ it appears convergence limiting singularity now just about as strong but for negative γ , so suggest we try $E = \frac{1}{3}$. Are the small improvements

worthwhile? Bit hard to say as I do not have Wannier interactions for the piecewise quadratic periodic potential.

We lose the link to classic consistency of finite differences, but do we care?

Probably the Euler parameter should depend upon the potential strength, but how?

Fin.

124 end;

Acknowledgement thanks to CSU and AMSI.

References

- Alfimov, G. L., Kevrekidis, P. G., Konotop, V. V. & Salerno, M. (2002), ‘Wannier functions analysis of the nonlinear Schrödinger equation with a periodic potential’, *Phys. Rev. E* **66**(046608), 1–6.
- Van Dyke, M. (1964), ‘Higher approximations in boundary-layer theory. Part 3. parabola in uniform stream’, *J. Fluid Mech.* **19**, 145–159.