Ensure continuity and self-adjointness of holistic discretisation of (n)lS equation with periodic potential wells

AJR

January 16, 2014

Execute in Reduce with `in_tex "potwell.tex"$`

Instead of the nlS, start by modelling the 1D diffusion PDE with potential wells $V(x) := -\frac{A\pi^2}{H^2}\cos(2\pi x/H)$,

$$-i\frac{\partial u}{\partial t} = \frac{\partial^2 u}{\partial x^2} - V(x)u$$

on a macroscale grid with same spacing as the wells: that is, locate the grid at the minimum of the wells at $X_j = jH$ say. The jth element is $X_{j-1} \le x \le X_j$.

Figure 2 plots subgrid fields for potential strength $A = 1$ which look good, but $A = 2, 3$ are ugly.

Improve printing.

```
on div; off allfac; on revpri;
factor hh,uu,d,aa;
```



Figure 1: beautiful C^2, piecewise quintic, approximation to the subgrid fields of the 'diffusion' PDE, $A=0$, with coupling errors $\mathcal{O}(\gamma^3)$: red, $0\leq\xi\leq 1$; green, $1\leq\xi\leq 2$; blue, $2\leq\xi\leq 3$; magenta, $3\leq\xi\leq 4$.

Define shift right/left operators `ep` and `em`: use that in terms of centred mean and difference operators, μ and δ, they are $1\pm\mu\delta+\frac{1}{2}\delta^2$ (**?**, p.65). Also encode the identity that $\mu^2=1+\delta^2/4$. Define the 'spline' operator `ss` $=S:=(1+\delta^2/6)^{-1}$. Have verified that, on an infinite lattice, operator S is the Toeplitz matrix with elements

$$S_{ij}=\sqrt{3}(-2+\sqrt{3})^{|i-j|}=\sqrt{3}(-0.2679)^{|i-j|}.$$

```
ep:=1+mu*del+del^2/2;
```



Figure 2: beautiful C^2 approximation to the subgrid fields of the PDE with potential strength $A = 1$ and coupling errors $\mathcal{O}(\gamma^2)$ (should only be C^1!): red, $0 \le \xi \le 1$; green, $1 \le \xi \le 2$; blue, $2 \le \xi \le 3$; magenta, $3 \le \xi \le 4$.

```
em:=1-mu*del+del^2/2;
let { mu^2=>1+del^2/4
    , ss*del^2=>6-6*ss };
```

Write the solution in terms of the microscale variable $\xi := (x - X_{j-1})/H$.

```
depend xi,x;
let df(~a,x)=>df(a,xi)/hh;
```

To find corrections, linear operator `linv` solves DEs of the form $\partial^2\hat u/\partial\xi^2 = \text{Res}$ such that $\hat u = 0$ at $\xi = 0, 1$.

```
operator linv; linear linv;
let { linv(xi^~~p,xi)=>(xi^(p+2)-xi)/(p+1)/(p+2)
    , linv(1,xi)=>(xi^2-xi)/2
    , linv(cos(~~m*pi*xi),xi)
        =>(1-cos(m*pi*xi))/(m*pi)^2 when evenp(m)
    , linv(sin(~~m*pi*xi),xi)=>-sin(m*pi*xi)/(m*pi)^2
    , linv(xi^~~p*cos(~~m*pi*xi),xi)
        =>xi^p*(1-cos(m*pi*xi))/(m*pi)^2 when evenp(m)
    , linv(xi^~~p*sin(~~m*pi*xi),xi)
        =>-xi^p*sin(m*pi*xi)/(m*pi)^2
    };
```

Write the slow manifold in terms of amplitudes $U_j(t) := u(X_j, t)$. These depend upon time according to $dU_j/dt = g_j$. We let all the j dependence be in the operators.

```
depend uu,t;
let df(uu,t)=>g;
```

The linear solution are equilibria, $g = 0$, of piecewise linear field between U_{j-1} at $\xi = 0$ and U_j at $\xi = 1$.

```
g:=0;
u:=xi*uu+(1-xi)*em*uu;
```

Iterate until the slow manifold model is found to the following specified order of accuracy.

```
let { gamma^2=>0, aa^2=>0 };
for it:=1:99 do begin
```

Compute residuals of governing equations.

```
    pde:=trigsimp(
        i*df(u,t)+df(u,x,x)+aa*pi^2/hh^2*cos(2*pi*xi)*u
        , combine);
    amp:=sub(xi=1,u)-uu;
    cty:=sub(xi=0,ep*u)-sub(xi=1,u);
    hux:=hh*df(u,x)$
    jmp:=-sub(xi=0,ep*hux)+sub(xi=1,hux)
        +(1-gamma)*sub(xi=1,ep*u-2*u+em*u);
    write lengthRes:=map(length(~a),{pde,amp,cty,jmp});
```

Correct approximations based upon the residuals. These ad hoc corrections are not optimal, but they do work after enough iterations.

```
    g:=g+i*(gd:=-ss*jmp/hh^2);
    u:=u-linv(pde-(xi+(1-xi)*em)*gd,xi)*hh^2;
```

Exit the loop when all residuals are zero to the order specified.

```
    showtime;
    if {pde,amp,cty,jmp}={0,0,0,0} then write it:=it+10000;
end;
```

Confirm left-right symmetry.

```
v:=trigsimp(sub({del=-del,xi=1-xi},ep*u))$
duv:=trigsimp(v-u);
```

Get equivalent PDE, but need to improve to be able to analyse to any order. Here, `d` denotes $\partial/\partial x$.

```
mig:=-i*g;
```

```
ssd:=1-hh^2*d^2/6+hh^4*d^4/72-hh^6*d^6/2160;
let d^7=>0;
migde:=sub(ss=ssd,-i*g);
```

Optionally plot some fully coupled subgrid fields of the linear problem, $\epsilon = 0$, for a potential strength $A = 1$, say. Result is beautiful.

```
load_package gnuplot;
if 1 then begin
hh:=1; gamma:=1; epsilon:=0;
u:=(sub(mu=delmu/del,u) where del^2=>delsq);
off nat;out "tmp.red"$
write u:=u$
write g:=g$
write "end;"$
shut "tmp.red"$ on nat;
n:=20; % 20 may be enough, but increase to test
matrix ss(2*n-1,2*n-1),delsq(2*n-1,2*n-1),delmu(2*n-1,2*n-1),u
for i:=1:2*n-1 do
    for j:=1:2*n-1 do
        ss(i,j):=sqrt(3)*(-2+sqrt(3))^abs(i-j);
for i:=1:2*n-1 do uu(i,i):=1;
for i:=1:2*n-1 do delsq(i,i):=-2;
for i:=2:2*n-1 do delsq(i,i-1):=delsq(i-1,i):=1;
for i:=2:2*n-1 do delmu(i,i-1):=-1/2;
for i:=2:2*n-1 do delmu(i-1,i):=+1/2;
on rounded; print_precision 4;
in "tmp.red";
write aa:=1;
ujs:=map(max(-1/2,min(3/2,~a)),
      {u(n,n)
      ,sub(xi=xi-1,u(n+1,n))
      ,sub(xi=xi-2,u(n+2,n))
```

```
      ,sub(xi=xi-3,u(n+3,n))
      })$
plot(ujs,xi=(0 .. 4));
end;
```

Fin.

```
end;
```