

Ensure continuity and self-adjointness of holistic discretisation of (n)IS equation with periodic potential wells

AJR

January 16, 2014

Execute in Reduce with `in_tex "potwell.tex"$`

Instead of the nIS, start by modelling the 1D diffusion PDE with potential wells $V(x) := -\frac{A\pi^2}{H^2} \cos(2\pi x/H)$,

$$-i\frac{\partial u}{\partial t} = \frac{\partial^2 u}{\partial x^2} - V(x)u$$

on a macroscale grid with same spacing as the wells: that is, locate the grid at the minimum of the wells at $X_j = jH$ say. The j th element is $X_{j-1} \leq x \leq X_j$.

Improve printing.

```
1 on div; off allfac; on revpri;  
2 factor hh,uu,d,aa;
```

Define shift right/left operators **ep** and **em**: use that in terms of centred mean and difference operators, μ and δ , they are $1 \pm \mu\delta + \frac{1}{2}\delta^2$ (? , p.65). Also encode the identity that $\mu^2 = 1 + \delta^2/4$. Define the ‘spline’ operator




Figure 1: beautiful C^2 , piecewise quintic, approximation to the subgrid fields of the ‘diffusion’ PDE, $A = 0$, with coupling errors $\mathcal{O}(\gamma^3)$: red, $0 \leq \xi \leq 1$; green, $1 \leq \xi \leq 2$; blue, $2 \leq \xi \leq 3$; magenta, $3 \leq \xi \leq 4$.

$\mathbf{ss} = S := (1 + \delta^2/6)^{-1}$. Have verified that, on an infinite lattice, operator S is the Toeplitz matrix with elements

$$S_{ij} = \sqrt{3}(-2 + \sqrt{3})^{|i-j|} = \sqrt{3}(-0.2679)^{|i-j|}.$$

```

3 ep:=1+mu*del+del^2/2;
4 em:=1-mu*del+del^2/2;
5 let { mu^2=>1+del^2/4
6      , ss*del^2=>6-6*ss };

```




Figure 2: beautiful C^2 approximation to the subgrid fields of the PDE with potential strength $A = 1$ and coupling errors $\mathcal{O}(\gamma^2)$ (should only be C^1 !): red, $0 \leq \xi \leq 1$; green, $1 \leq \xi \leq 2$; blue, $2 \leq \xi \leq 3$; magenta, $3 \leq \xi \leq 4$.

Write the solution in terms of the microscale variable $\xi := (x - X_{j-1})/H$.

```
7 depend xi,x;
8 let df(~a,x)=>df(a,xi)/hh;
```

To find corrections, linear operator **linv** solves DEs of the form $\partial^2 \hat{u} / \partial \xi^2 = \text{Res}$ such that $\hat{u} = 0$ at $\xi = 0, 1$.

```
9 operator linv; linear linv;
```

```

10 let { linv(xi^^~p,xi)=>(xi^(p+2)-xi)/(p+1)/(p+2)
11      , linv(1,xi)=>(xi^2-xi)/2
12      , linv(cos(~~m*pi*xi),xi)
13          =>(1-cos(m*pi*xi))/(m*pi)^2 when evenp(m)
14      , linv(sin(~~m*pi*xi),xi)=>-sin(m*pi*xi)/(m*pi)^2
15      , linv(xi^^~p*cos(~~m*pi*xi),xi)
16          =>xi^p*(1-cos(m*pi*xi))/(m*pi)^2 when evenp(m)
17      , linv(xi^^~p*sin(~~m*pi*xi),xi)
18          =>-xi^p*sin(m*pi*xi)/(m*pi)^2
19      };

```

Write the slow manifold in terms of amplitudes $U_j(t) := u(X_j, t)$. These depend upon time according to $dU_j/dt = g_j$. We let all the j dependence be in the operators.

```

20 depend uu,t;
21 let df(uu,t)=>g;

```

The linear solution are equilibria, $g = 0$, of piecewise linear field between U_{j-1} at $\xi = 0$ and U_j at $\xi = 1$.

```

22 g:=0;
23 u:=xi*uu+(1-xi)*em*uu;

```

Iterate until the slow manifold model is found to the following specified order of accuracy.

```

24 let { gamma^2=>0, aa^2=>0 };
25 for it:=1:99 do begin

```

Compute residuals of governing equations.

```

26     pde:=trigsimp(

```

```

27      i*df(u,t)+df(u,x,x)+aa*pi^2/hh^2*cos(2*pi*xi)*u
28      , combine);
29      amp:=sub(xi=1,u)-uu;
30      cty:=sub(xi=0,ep*u)-sub(xi=1,u);
31      hux:=hh*df(u,x)$
32      jmp:=-sub(xi=0,ep*hux)+sub(xi=1,hux)
33      +(1-gamma)*sub(xi=1,ep*u-2*u+em*u);
34      write lengthRes:=map(length(~a),{pde,amp,cty,jmp});

```

Correct approximations based upon the residuals. These ad hoc corrections are not optimal, but they do work after enough iterations.

```

35      g:=g+i*(gd:=-ss*jmp/hh^2);
36      u:=u-linv(pde-(xi+(1-xi)*em)*gd,xi)*hh^2;

```

Exit the loop when all residuals are zero to the order specified.

```

37      showtime;
38      if {pde,amp,cty,jmp}={0,0,0,0} then write it:=it+10000;
39 end;

```

Confirm left-right symmetry.

```

40 v:=trigsimp(sub({del=-del,xi=1-xi},ep*u))$
41 duv:=trigsimp(v-u);

```

Get equivalent PDE, but need to improve to be able to analyse to any order. Here, d denotes $\partial/\partial x$.

```

42 mig:=-i*g;
43 ssd:=1-hh^2*d^2/6+hh^4*d^4/72-hh^6*d^6/2160;
44 let d^7=>0;
45 migde:=sub(ss=ssd,-i*g);

```

Optionally plot some fully coupled subgrid fields of the linear problem, $\epsilon = 0$, for a potential strength $A = 1$, say. Result is beautiful.

```

46 load_package gnuplot;
47 if 1 then begin
48 hh:=1; gamma:=1; epsilon:=0; aa:=1;
49 u:=(sub(mu=delmu/del,u) where del^2=>delsq);
50 off nat;out "tmp.red"$
51 write u:=u$
52 write g:=g$
53 write "end;"$
54 shut "tmp.red"$ on nat;
55 n:=20; % 20 may be enough, but increase to test
56 matrix ss(2*n-1,2*n-1),delsq(2*n-1,2*n-1),delmu(2*n-1,2*n-1),u
57 for i:=1:2*n-1 do
58     for j:=1:2*n-1 do
59         ss(i,j):=sqrt(3)*(-2+sqrt(3))^abs(i-j);
60 for i:=1:2*n-1 do uu(i,i):=1;
61 for i:=1:2*n-1 do delsq(i,i):=-2;
62 for i:=2:2*n-1 do delsq(i,i-1):=delsq(i-1,i):=1;
63 for i:=2:2*n-1 do delmu(i,i-1):=-1/2;
64 for i:=2:2*n-1 do delmu(i-1,i):=+1/2;
65 on rounded; print_precision 4;
66 in "tmp.red";
67 ujs:=map(max(-1/2,min(3/2,~a)),
68     {u(n,n)
69     ,sub(xi=xi-1,u(n+1,n))
70     ,sub(xi=xi-2,u(n+2,n))
71     ,sub(xi=xi-3,u(n+3,n))
72     });
73 plot(ujs,xi=(0 .. 4));
74 end;

```

Fin.

75 end;